
    Getting Started with Cucumber for Elixir

Cucumber for Elixir allows you to write executable specifications using natural language that bridges the gap between technical and non-technical stakeholders.
Overview
The Cucumber framework allows you to write tests in a natural language format using Gherkin syntax. This approach enables:
	Behavior-driven development (BDD)
	Clear documentation of application behavior
	Tests that serve as living documentation
	Auto-discovery of feature files and step definitions

Installation
Add cucumber to your mix.exs dependencies:
def deps do
  [
    {:cucumber, "~> 0.4.0"}
  ]
end
Quick Start
1. Configure Test Helper
Add Cucumber to your test/test_helper.exs:
ExUnit.start()
Cucumber.compile_features!()
2. Create a Feature File
Feature files use the Gherkin syntax and should be placed in test/features/ with a .feature extension.
# test/features/user_authentication.feature
Feature: User Authentication

Background:
  Given the application is running

Scenario: User signs in with valid credentials
  Given I am on the sign in page
  When I enter "user@example.com" as my email
  And I enter "password123" as my password
  And I click the "Sign In" button
  Then I should be redirected to the dashboard
  And I should see "Welcome back" message
3. Create Step Definitions
Create step definitions in test/features/step_definitions/ with a .exs extension:
# test/features/step_definitions/authentication_steps.exs
defmodule AuthenticationSteps do
  use Cucumber.StepDefinition
  import ExUnit.Assertions

  # Step definitions
  step "the application is running" do
    # Setup code here
    %{app_started: true}
  end

  step "I am on the sign in page", context do
    # Navigate to sign in page
    Map.put(context, :current_page, :sign_in)
  end

  step "I enter {string} as my email", %{args: [email]} = context do
    # Code to enter email
    Map.put(context, :email, email)
  end

  step "I enter {string} as my password", %{args: [password]} = context do
    # Code to enter password
    Map.put(context, :password, password)
  end

  step "I click the {string} button", %{args: [button_text]} do
    # Code to click button
    {:ok, %{clicked: button_text}}
  end

  step "I should be redirected to the dashboard", context do
    # Assertions for redirection
    assert context.current_page == :dashboard
    context
  end

  step "I should see {string} message", %{args: [message]} do
    # Assertion for message
    assert_text(message)
    :ok
  end
end
4. Run Your Tests
Run your tests using the standard mix test command:
# Run all tests including Cucumber
mix test

# Run only Cucumber tests
mix test --only cucumber

# Run tests for a specific feature
mix test --only feature_user_authentication

File Structure
Cucumber automatically discovers features and steps using this structure:
test/
  features/                      # Feature files
    user_authentication.feature
    shopping_cart.feature
    step_definitions/           # Step definition files
      authentication_steps.exs
      shopping_steps.exs
      common_steps.exs
Async Test Execution
For features that don't share state, you can enable concurrent test execution by adding the @async tag:
# test/features/calculator.feature
@async
Feature: Calculator Operations
  Independent calculations that can run concurrently

Scenario: Addition
  Given I have numbers 10 and 20
  When I add them together
  Then the result should be 30

Scenario: Multiplication
  Given I have numbers 5 and 6
  When I multiply them
  Then the result should be 30
Async features run concurrently with other async tests, improving test suite performance. Only use @async for features that are truly independent and don't rely on test execution order.
When using Ecto, async tests work well with the SQL sandbox in shared mode. See your application's test setup for database configuration.
Next Steps
For more detailed information about the Cucumber for Elixir framework, check out these guides:
	Feature Files - Learn how to write feature files with Gherkin syntax
	Step Definitions - Learn how to implement step definitions
	Error Handling - Understanding error reporting and debugging
	Best Practices - Best practices and examples



  

    Writing Feature Files

Feature files are the heart of Cucumber testing. They're written in Gherkin syntax, a business-readable domain-specific language that lets you describe application behavior without detailing how that behavior is implemented.
Feature File Structure
Feature files consist of several components:
Feature
Every feature file starts with the Feature: keyword followed by a name and optional description:
Feature: Shopping Cart
  As a user
  I want to add items to my cart
  So that I can purchase them later
Background
The Background: section contains steps that are executed before each scenario:
Background:
  Given I am logged in as a customer
  And the product catalog is available
Scenarios
Scenarios are concrete examples of how the feature should behave:
Scenario: Adding an item to an empty cart
  Given I am on the product page for "Ergonomic Keyboard"
  When I click "Add to Cart"
  Then I should see "Item added to cart" message
  And my cart should contain 1 item
Steps
Steps use keywords like Given, When, Then, And, and But:
	Given: Establishes preconditions
	When: Describes actions
	Then: Specifies expected outcomes
	And/But: Continues the previous step type

Step Arguments
Gherkin supports several types of step arguments:
Data Tables
Scenario: Adding multiple items to cart
  Given I have the following items in my cart:
    | Product Name    | Quantity | Price |
    | Smartphone      | 1        | 699.99|
    | Protection Plan | 1        | 79.99 |
  When I proceed to checkout
  Then the total should be 779.98
Doc Strings
Scenario: Submit feedback
  When I submit the following feedback:
    """
    I really like your product, but I think
    it could be improved by adding more features.
    Keep up the good work!
    """
  Then my feedback should be recorded
File Organization
Feature files should be placed in a test/features/ directory and have a .feature extension. Organize them logically by feature or domain area:
test/
└── features/
    ├── authentication/
    │   ├── login.feature
    │   └── registration.feature
    ├── shopping/
    │   ├── cart.feature
    │   └── checkout.feature
    └── user_profile.feature
Tags
Tags are used to categorize scenarios and can be used for filtering which tests to run:
@authentication @important
Feature: User Login

@happy_path
Scenario: Successful login with valid credentials
  Given I am on the login page
  When I enter valid credentials
  Then I should be logged in

@error_handling
Scenario: Failed login with invalid credentials
  Given I am on the login page
  When I enter invalid credentials
  Then I should see an error message
Tags can be used at the Feature level (applies to all scenarios) or at the Scenario level.
Special Tags
@async
The @async tag enables concurrent test execution for features that don't share state:
@async
Feature: Independent Calculator Operations
  This feature can run concurrently with other async tests

Scenario: Addition
  Given I have a calculator
  When I add 5 and 3
  Then the result should be 8

Scenario: Multiplication
  Given I have a calculator
  When I multiply 4 and 7
  Then the result should be 28
Use @async only for features that:
	Don't share state with other tests
	Don't rely on test execution order
	Are truly independent of other tests

Note: Database tests can safely run async when using Ecto's SQL sandbox in shared mode. For non-Ecto resources (files, external APIs, etc.), ensure they can handle concurrent access.
Best Practices for Feature Files
	Keep language simple and consistent - Use clear, business-focused terminology
	One scenario per behavior - Each scenario should test one specific behavior
	Be specific in examples - Use concrete values rather than vague descriptions
	Use background wisely - Only for steps that are truly common to all scenarios
	Limit scenario length - If a scenario has many steps, consider breaking it down
	Organize with tags - Use tags to categorize and group related scenarios
	Think like a user - Write scenarios from the user's perspective
	Don't get too technical - Avoid technical implementation details in Gherkin



  

    Step Definitions

Step definitions connect the Gherkin steps in your feature files to actual code. They're the glue between your natural language specifications and the implementation that tests your application.
Creating Step Definition Files
Step definitions should be placed in test/features/step_definitions/ with a .exs extension:
# test/features/step_definitions/authentication_steps.exs
defmodule AuthenticationSteps do
  use Cucumber.StepDefinition
  import ExUnit.Assertions

  # Step definitions go here
end
Basic Step Definition
Step definitions are created using the step macro:
step "I am logged in as a customer", context do
  # Authentication logic here
  Map.put(context, :user, create_and_login_customer())
end
Steps with Parameters
Cucumber supports several parameter types that can be used in step patterns. Parameters are accessed through pattern matching on the args field of the context:
String Parameters
step "I am on the product page for {string}", %{args: [product_name]} do
  # Navigate to product page
  %{current_page: :product, product_name: product_name}
end
Integer Parameters
step "I should have {int} items in my wishlist", %{args: [expected_count]} = context do
  # Assertion for wishlist count
  assert get_wishlist_count(context) == expected_count
  context
end
Float Parameters
step "the total price should be {float}", %{args: [expected_total]} = context do
  # Assertion for price
  assert_in_delta get_cart_total(context), expected_total, 0.01
  context
end
Word Parameters
step "I should see the {word} dashboard", %{args: [dashboard_type]} = context do
  # Assertion for dashboard type
  assert get_current_dashboard(context) == dashboard_type
  context
end
Multiple Parameters
When a step has multiple parameters, you can pattern match on all of them:
step "I transfer {float} from {string} to {string}", %{args: [amount, from_account, to_account]} do
  # Transfer logic
  %{transfer: %{amount: amount, from: from_account, to: to_account}}
end
Working with Data Tables
In your feature file:
Given I have the following items in my cart:
  | Product Name    | Quantity | Price |
  | Smartphone      | 1        | 699.99|
  | Protection Plan | 1        | 79.99 |
In your step definitions:
step "I have the following items in my cart:", context do
  # Access the datatable
  datatable = context.datatable

  # Access headers
  headers = datatable.headers  # ["Product Name", "Quantity", "Price"]

  # Access rows as maps
  items = datatable.maps
  # [
  #   %{"Product Name" => "Smartphone", "Quantity" => "1", "Price" => "699.99"},
  #   %{"Product Name" => "Protection Plan", "Quantity" => "1", "Price" => "79.99"}
  # ]

  # Process the items
  Map.put(context, :cart_items, items)
end
Working with DocStrings
DocStrings allow you to pass multi-line text to a step:
In your feature file:
When I submit the following JSON:
  """
  {
    "name": "Test Product",
    "price": 29.99,
    "available": true
  }
  """
In your step definitions:
step "I submit the following JSON:", context do
  # The docstring is available in context.docstring
  json_data = Jason.decode!(context.docstring)

  # Process the JSON
  Map.put(context, :submitted_data, json_data)
end
Return Values
Step definitions must return one of the following values (matching ExUnit's setup behavior):
	:ok - Keeps the context unchanged
	A map - Merged into the existing context
	A keyword list - Merged into the existing context
	{:ok, map_or_keyword_list} - Merged into the existing context
	{:error, reason} - Fails the step with the given reason

Reusable Step Definitions
You can create reusable step definitions that can be shared across multiple features:
# test/features/step_definitions/common_steps.exs
defmodule CommonSteps do
  use Cucumber.StepDefinition
  import ExUnit.Assertions

  step "I wait {int} seconds", %{args: [seconds]} = context do
    Process.sleep(seconds * 1000)
    context
  end

  step "I should see {string}", %{args: [text]} = context do
    assert page_contains_text?(context, text)
    context
  end
end
Best Practices
	Keep steps focused: Each step should do one thing well
	Use descriptive step patterns: Make your steps readable and self-documenting
	Share common steps: Create reusable step definitions for common actions
	Handle errors gracefully: Return {:error, reason} for expected failures
	Maintain context: Always return the context (or :ok) to maintain state between steps



  

    Error Handling and Debugging

Cucumber for Elixir provides detailed error reporting to help debug test failures. This document explains how to handle and understand errors in your Cucumber tests.
Types of Errors
Missing Step Definition
When a step in your feature file has no matching definition, the framework will provide a helpful error message:
** (Cucumber.StepError) No matching step definition found for step:

  When I try to use a step with no definition

in scenario "Missing Step Example" (test/features/example.feature:6)

Please define this step with:

step "I try to use a step with no definition", context do
  # Your step implementation here
  context
end
This error not only tells you which step is missing, but also provides a template for implementing the missing step definition.
Failed Step
When a step fails during execution (due to a failed assertion or an exception), you'll get an error like this:
** (Cucumber.StepError) Step failed:

  Then the validation should succeed

in scenario "Form Submission" (test/features/forms.feature:12)
matching pattern: "the validation should succeed"

Validation failed: invalid input data

Step execution history:
  [passed] Given a form to fill out
  [passed] When I submit invalid data
  [failed] Then the validation should succeed
The error message includes:
	Which step failed
	The scenario and file where the failure occurred
	The step pattern that matched
	The error message from the step implementation
	The execution history, showing which steps passed and which failed

Syntax Errors in Feature Files
If there are syntax errors in your feature files, you'll get an error when the parser tries to process the file:
** (Cucumber.ParseError) Syntax error in feature file:

  test/features/invalid.feature:5

Expected a scenario or background but found:

  Invalid line that doesn't start with a Gherkin keyword
Step Failure Handling
Step definitions can indicate failure in several ways:
1. Assertions
Using ExUnit assertions will cause the step to fail if the assertion fails:
step "the total should be {float}", %{args: [total]} = context do
  assert context.cart_total == total
  :ok
end
2. Raising Exceptions
Any uncaught exception will cause the step to fail:
step "I click the submit button", _context do
  raise "The submit button is disabled"
  :ok
end
3. Returning {:error, reason}
You can explicitly return an error tuple to fail a step:
step "the payment should be successful", context do
  if context.payment_status == :success do
    :ok
  else
    {:error, "Expected payment to succeed, but got status: #{context.payment_status}"}
  end
end
Debugging Tips
1. Use IO.inspect for Debug Output
Insert IO.inspect calls to see the values of variables during test execution:
step "I should see my order summary", context do
  IO.inspect(context, label: "Context in order summary step")
  assert context.order != nil
  :ok
end
2. Add Step Execution Logs
Log information about step execution:
step "I complete the checkout process", context do
  IO.puts("Starting checkout process")
  # Checkout logic
  IO.puts("Completed checkout process")
  :ok
end
3. Examine the Full Context
Print the full context at any point to see the accumulated state:
step "I check my context", context do
  IO.inspect(context, label: "Current context", pretty: true)
  :ok
end
4. Create Debug-Only Steps
Add steps specifically for debugging:
step "I debug my test state", context do
  IO.puts("==== DEBUG STATE ====")
  IO.inspect(context.current_page, label: "Current Page")
  IO.inspect(context.user, label: "Current User")
  IO.inspect(context.cart_items, label: "Cart Items")
  IO.puts("==== END DEBUG ====")
  :ok
end
Handling Flaky Tests
Sometimes tests can be inconsistent due to timing issues, especially with UI interactions:
step "I should see the confirmation message", context do
  # Add delay or retry logic for UI-related assertions
  :timer.sleep(500)  # Give the UI time to update
  assert_text("Your order has been confirmed")
  :ok
end
Consider implementing retry logic for flaky steps:
defp retry_until(function, max_attempts \\ 5, delay \\ 100) do
  Enum.reduce_while(1..max_attempts, nil, fn attempt, _acc ->
    case function.() do
      {:ok, result} -> {:halt, {:ok, result}}
      {:error, reason} ->
        if attempt == max_attempts do
          {:halt, {:error, reason}}
        else
          :timer.sleep(delay)
          {:cont, nil}
        end
    end
  end)
end

step "I should see the success notification", _context do
  result = retry_until(fn ->
    if element_visible?(".success-notification") do
      {:ok, true}
    else
      {:error, "Success notification not visible"}
    end
  end)

  case result do
    {:ok, _} -> :ok
    {:error, reason} -> {:error, reason}
  end
end
Common Error Patterns and Solutions
	Error Pattern	Possible Cause	Solution
	Step not found	Step definition missing or typo in step	Create the missing step definition or correct the typo
	Assertion failure	Expected value doesn't match actual	Check your test data and application logic
	Timeout	Asynchronous operation didn't complete in time	Increase timeout or add retry logic
	Element not found	UI element not rendered yet or selector wrong	Add delay, retry, or correct the selector
	Context key missing	Previous step didn't set expected data	Ensure required context keys are set in earlier steps



  

    Best Practices for Cucumber Tests

This guide outlines best practices for writing and organizing your Cucumber tests to ensure they remain maintainable, readable, and effective.
Feature File Organization
Directory Structure
test/
├── features/                    # Feature files
│   ├── authentication/          # Feature grouping by domain
│   │   ├── login.feature
│   │   └── registration.feature
│   ├── shopping/               # Another domain
│   │   ├── cart.feature
│   │   └── checkout.feature
│   └── step_definitions/       # Step definition files
│       ├── authentication_steps.exs
│       ├── shopping_steps.exs
│       └── common_steps.exs
Naming Conventions
	Use snake_case for feature file names
	Group related features into subdirectories
	Name step definition modules descriptively (e.g., AuthenticationSteps, ShoppingSteps)
	Use .exs extension for step definition files

Writing Good Scenarios
Scenario Best Practices
	Keep scenarios focused: Each scenario should test one specific behavior
	Be consistent: Use consistent language across scenarios
	Use concrete examples: Prefer specific, realistic values to abstract placeholders
	Avoid technical details: Keep scenarios in business language
	Keep them short: Aim for 3-7 steps per scenario
	Use backgrounds wisely: Only for truly common setup steps

Example: Bad vs. Good
Bad:
Scenario: User interaction
  Given a user
  When the user does stuff
  Then the outcome is good
Good:
Scenario: Customer adds product to shopping cart
  Given I am logged in as "alice@example.com"
  And I am viewing the "iPhone 15 Pro" product page
  When I click "Add to Cart"
  Then I should see "iPhone 15 Pro" in my shopping cart
  And the cart total should be $999.00
Step Definition Best Practices
Keep Steps Reusable
Write generic steps that can be reused across features:
# Good - reusable
step "I click {string}", %{args: [button_text]} = context do
  click_button(context, button_text)
end

# Less reusable - too specific
step "I click the red submit button on the login form", context do
  click_specific_button(context)
end
Use the Context Effectively
Pass data between steps using the context:
step "I create a product named {string}", %{args: [name]} = context do
  product = create_product(name: name)
  Map.put(context, :product, product)
end

step "I should see the product in my catalog", context do
  assert product_in_catalog?(context.product)
  context
end
Organize Step Definitions
Group related steps in the same module:
# test/features/step_definitions/authentication_steps.exs
defmodule AuthenticationSteps do
  use Cucumber.StepDefinition
  import ExUnit.Assertions

  # Login steps
  step "I am logged in as {string}", %{args: [email]} = context do
    user = login_user(email)
    Map.put(context, :current_user, user)
  end

  # Registration steps
  step "I register with email {string}", %{args: [email]} = context do
    user = register_user(email: email)
    Map.put(context, :new_user, user)
  end
end
Common Patterns
Data Setup Pattern
Create helper functions for common data setup:
defmodule TestHelpers do
  def create_user(attrs \\ %{}) do
    default_attrs = %{
      email: "test@example.com",
      name: "Test User"
    }

    attrs = Map.merge(default_attrs, attrs)
    # Create user logic
  end
end

# In your steps
step "a user exists with email {string}", %{args: [email]} do
  user = TestHelpers.create_user(email: email)
  %{user: user}
end
Assertion Helpers
Create custom assertion helpers for cleaner steps:
defmodule AssertionHelpers do
  import ExUnit.Assertions

  def assert_logged_in(context) do
    assert context[:current_user] != nil
    assert context[:session_token] != nil
  end

  def assert_product_visible(context, product_name) do
    assert product_name in get_visible_products(context)
  end
end
Testing Tips
Use Tags for Organization
Tag your scenarios for easy filtering:
@authentication @smoke
Scenario: Successful login
  Given I am on the login page
  When I enter valid credentials
  Then I should be logged in

@wip @slow
Scenario: Complex data processing
  Given a large dataset
  When I process the data
  Then the results should be accurate
Run specific tags:
mix test --only authentication
mix test --exclude wip

Background vs. Helper Steps
Use backgrounds for truly common setup that applies to all scenarios:
Background:
  Given the system is initialized
  And default products exist

Scenario: View product catalog
  When I visit the catalog page
  Then I should see all products
Handling Asynchronous Operations
For operations that might take time:
step "I wait for the email to arrive", context do
  # Poll for the email with a timeout
  email = wait_for_email(context.current_user.email, timeout: 5_000)
  Map.put(context, :received_email, email)
end

defp wait_for_email(email, opts) do
  timeout = Keyword.get(opts, :timeout, 5_000)
  poll_interval = 100

  wait_until(timeout, poll_interval, fn ->
    check_email_arrived(email)
  end)
end
Debugging Tips
	Use IO.inspect in steps: Temporarily add IO.inspect(context) to see the current state
	Run single scenarios: Focus on one test at a time during debugging
	Check step history: Error messages show which steps ran before failure
	Use meaningful assertions: Include context in assertion messages

step "the order should be completed", context do
  order = context.order
  assert order.status == "completed",
         "Expected order #{order.id} to be completed, but was #{order.status}"
  context
end


  

    Cucumber Architecture

This document provides an overview of the Cucumber implementation architecture, explaining the core components and how they interact.
Core Components
Cucumber
  ├── Discovery (Feature/Step Finding)
  ├── Gherkin (Parser)
  ├── Expression (Parameter Matching)
  ├── Compiler (Test Generation)
  ├── Runtime (Step Execution)
  └── StepError (Error Reporting)
Discovery System
The discovery system automatically finds and loads feature files and step definitions:
Discovery.discover()
  ├── Scans for features (test/features/**/*.feature)
  ├── Loads step definitions (test/features/step_definitions/**/*.exs)
  └── Returns: %{features: [...], step_registry: %{...}}
Gherkin Parser
The Gherkin parser is responsible for parsing .feature files into a structured format that can be executed. It handles the syntax of Gherkin, including:
	Feature declarations
	Scenario outlines
	Backgrounds
	Steps (Given, When, Then)
	Tables and doc strings
	Tags

The parser produces an Abstract Syntax Tree (AST) that represents the structure of the feature file.
# Simplified representation of the Gherkin parser flow
Feature File (Text) → Lexer → Tokens → Parser → AST
Expression Engine
The Expression engine is responsible for matching step text against step definitions. It supports:
	Regular expressions
	Cucumber expressions (a simplified syntax with parameter types)
	Parameter conversion (string to typed values)

defmodule Cucumber.Expression do
  # Converts a cucumber expression into a regex and parameter converters
  def compile(pattern) do
    # Transforms {string}, {int}, etc. into regex patterns
    # Returns {regex, converters}
  end

  # Matches text against a compiled expression
  def match(text, {regex, converters}) do
    # Returns {:match, args} or :no_match
  end
end
Compiler
The compiler generates ExUnit test modules from discovered features:
Compiler.compile_features!()
  ├── For each feature file:
  │   ├── Generates a test module
  │   ├── Creates setup from Background
  │   ├── Creates test cases from Scenarios
  │   └── Adds appropriate tags
  └── Compiles modules into memory
Runtime
The runtime executes steps during test runs:
Runtime.execute_step(context, step, step_registry)
  ├── Finds matching step definition
  ├── Prepares context with args, datatables, docstrings
  ├── Executes step function
  └── Processes return value
StepDefinition Macro
The StepDefinition module provides the DSL for defining steps:
defmodule MySteps do
  use Cucumber.StepDefinition

  step "pattern", context do
    # implementation
  end
end
Execution Flow
	Discovery Phase (at compile time)
	Cucumber.compile_features!() is called in test_helper.exs
	Discovery system finds all features and step definitions
	Step registry is built with pattern → module mappings


	Compilation Phase
	For each feature, a test module is generated
	Background steps become setup blocks
	Scenarios become test cases
	Tags are added for filtering


	Execution Phase (at runtime)
	ExUnit runs the generated test modules
	Each test executes its steps via Runtime
	Context is passed between steps
	Errors are reported with helpful messages



Data Flow
Feature File → Parser → AST
                          ↓
Step Files → Discovery → Registry
                          ↓
                      Compiler → Test Modules
                                      ↓
                                  ExUnit → Results
Key Design Decisions
Auto-Discovery
	Features and steps are automatically discovered
	No need to explicitly wire features to test modules
	Follows Ruby Cucumber's convention-over-configuration approach

ExUnit Integration
	Generated tests are standard ExUnit test modules
	Full support for ExUnit features (tags, setup, async)
	Works with existing test tooling

Runtime Compilation
	Tests are generated at runtime when mix test runs
	Allows for dynamic test generation
	No generated files to manage

Context Management
	ExUnit context is used directly
	Background steps modify context in setup
	Each step can read and modify context

Error Handling
Cucumber provides detailed error messages for common issues:
	Undefined Steps: Shows the exact step text and suggests implementation
	Step Failures: Shows the error, step location, and execution history
	Duplicate Steps: Detected at load time with file/line information

Extensibility
The architecture supports several extension points:
	Custom Parameter Types: Add new parameter types to Expression
	Custom Formatters: Create custom output formats
	Hooks: Before/after scenario hooks (via ExUnit setup/teardown)
	Step Libraries: Create reusable step definition modules



  

    
Cucumber 
    



      
A behavior-driven development (BDD) testing framework for Elixir using Gherkin syntax.
Cucumber is a testing framework that allows you to write executable specifications
in natural language. It bridges the gap between technical and non-technical stakeholders
by allowing tests to be written in plain language while being executed as code.
Setup
Add to your test_helper.exs:
Cucumber.compile_features!()
File Structure
By default, Cucumber expects the following structure:
test/
  features/
    authentication.feature
    shopping.feature
    step_definitions/
      authentication_steps.exs
      shopping_steps.exs
      common_steps.exs
    support/
      hooks.exs
Configuration
You can customize paths in config/test.exs:
config :cucumber,
  features: ["test/features/**/*.feature"],
  steps: ["test/features/step_definitions/**/*.exs"]
Step Definitions
Create step definition modules using Cucumber.StepDefinition:
defmodule AuthenticationSteps do
  use Cucumber.StepDefinition

  step "I am logged in as {string}", %{args: [username]} = context do
    {:ok, Map.put(context, :current_user, username)}
  end
end
Running Tests
Cucumber tests run with mix test and can be filtered using tags:
# Run all tests including Cucumber
mix test

# Run only Cucumber tests
mix test --only cucumber

# Exclude Cucumber tests
mix test --exclude cucumber
Key Features
	Auto-discovery of features and step definitions
	Integration with ExUnit's tagging system
	Context passing between steps
	Support for data tables and doc strings
	Rich error reporting with suggestions

        compile_features!(opts \\ [])

      


        Discovers and compiles all cucumber features into ExUnit tests.

Discovers and compiles all cucumber features into ExUnit tests.
This function should be called in your test_helper.exs file.
Options
	:features - List of patterns for feature files
	:steps - List of patterns for step definition files
	:support - List of patterns for support files

Examples
# Use default paths
Cucumber.compile_features!()

# Use custom paths
Cucumber.compile_features!(
  features: ["test/acceptance/**/*.feature"],
  steps: ["test/acceptance/steps/**/*.exs"]
)

  


        

      


  

    
Cucumber.Compiler 
    



      
Compiles discovered features and steps into ExUnit test modules.

      


      
        Summary


  
    Functions
  


    
      
        compile_features!(opts \\ [])

      


        Compiles all discovered features into ExUnit test modules.



    





      


      
        Functions

        


    

  
    
      
    
    
      compile_features!(opts \\ [])



        
          
        

    

  


  

Compiles all discovered features into ExUnit test modules.

  


        

      


  

    
Cucumber.Discovery 
    



      
Discovers and loads feature files and step definitions.

      


      
        Summary


  
    Functions
  


    
      
        discover(opts \\ [])

      


        Discovers all features and steps based on configuration.
Returns a struct containing parsed features and a registry of steps.

      discover(opts \\ [])



        
          
        

    

  


  

Discovers all features and steps based on configuration.
Returns a struct containing parsed features and a registry of steps.

  


        

      


  

    
Cucumber.Expression 
    



      
Parser and matcher for Cucumber Expressions used in step definitions.
Cucumber Expressions are a human-friendly alternative to regular expressions,
allowing you to define step patterns with typed parameters. This module handles
compiling these expressions to regular expressions and extracting/converting parameters.
Parameter Types
The following parameter types are supported:
	{string} - Matches quoted strings ("example") and converts to string
	{int} - Matches integers (42) and converts to integer
	{float} - Matches floating point numbers (3.14) and converts to float
	{word} - Matches a single word (no whitespace) and converts to string

Examples
# Matching a step with a string parameter
"I click {string} button" would match:
"I click "Submit" button"

# Matching a step with multiple parameters
"I add {int} items to my {word} list" would match:
"I add 5 items to my shopping list"

      


      
        Summary


  
    Functions
  


    
      
        compile(pattern)

      


        Compiles a Cucumber Expression pattern into a regex and parameter converters.



    


    
      
        match(text, arg)

      


        Matches a step text against a compiled Cucumber Expression.



    





      


      
        Functions

        


  
    
      
    
    
      compile(pattern)



        
          
        

    

  


  

Compiles a Cucumber Expression pattern into a regex and parameter converters.
This function transforms a human-readable pattern with typed parameters
into a regular expression for matching and a list of converter functions
to transform the captured values to their appropriate types.
Parameters
	pattern - A string containing a Cucumber Expression pattern

Returns
Returns a tuple {regex, converters} where:
	regex - A compiled regular expression for matching step text
	converters - A list of functions to convert captured values to their appropriate types

Examples
iex> {regex, converters} = Cucumber.Expression.compile("I have {int} items")
iex> Regex.match?(regex, "I have 42 items")
true

Matches a step text against a compiled Cucumber Expression.
This function attempts to match step text against a compiled Cucumber Expression
and extracts/converts any parameters if there's a match.
Parameters
	text - The step text to match against the pattern
	{regex, converters} - A compiled Cucumber Expression from compile/1

Returns
Returns one of:
	{:match, args} - If the text matches, where args is a list of converted parameter values
	:no_match - If the text doesn't match the expression

Examples
iex> compiled = Cucumber.Expression.compile("I have {int} items")
iex> Cucumber.Expression.match("I have 42 items", compiled)
{:match, [42]}
iex> Cucumber.Expression.match("I have no items", compiled)
:no_match

  


        

      


  

    
Cucumber.Hooks 
    



      
Provides hooks for setup and teardown in Cucumber tests.
Hooks can be defined globally or filtered by tags. They are executed
in the order they are defined, with Before hooks running in definition
order and After hooks running in reverse order.
Examples
defmodule DatabaseSupport do
  use Cucumber.Hooks

  # Global before hook - runs for all scenarios
  before_scenario context do
    # Setup code
    {:ok, Map.put(context, :setup, true)}
  end

  # Tagged before hook - only runs for @database scenarios
  before_scenario "@database", context do
    :ok = Ecto.Adapters.SQL.Sandbox.checkout(MyApp.Repo)

    if context.async do
      Ecto.Adapters.SQL.Sandbox.mode(MyApp.Repo, {:shared, self()})
    end

    {:ok, context}
  end

  # After hooks run in reverse order
  after_scenario _context do
    # Cleanup code
    :ok
  end
end

    Functions
  


    
      
        after_scenario(context_var, list)

      


        Defines an after_scenario hook that runs after each scenario.



    


    
      
        after_scenario(tag, context_var, list)

      


    


    
      
        before_scenario(context_var, list)

      


        Defines a before_scenario hook that runs before each scenario.



    


    
      
        before_scenario(tag, context_var, list)

      after_scenario(context_var, list)


        (macro)


        
          
        

    

  


  

Defines an after_scenario hook that runs after each scenario.
Can optionally be filtered by tag. After hooks run in reverse order
of definition. The hook receives the test context.

  



  
    
      
    
    
      after_scenario(tag, context_var, list)


        (macro)


        
          
        

    

  


  


  



  
    
      
    
    
      before_scenario(context_var, list)


        (macro)


        
          
        

    

  


  

Defines a before_scenario hook that runs before each scenario.
Can optionally be filtered by tag. The hook receives the test context
and must return one of:
	{:ok, context}
	:ok (keeps context unchanged)
	map (merged into context)


  



  
    
      
    
    
      before_scenario(tag, context_var, list)

Cucumber.Runtime 
    



      
Runtime execution of cucumber steps.

      


      
        Summary


  
    Functions
  


    
      
        execute_step(context, step, step_registry)

      


        Executes a step with the given context and step registry.

      execute_step(context, step, step_registry)



        
          
        

    

  


  

Executes a step with the given context and step registry.

  


        

      


  

    
Cucumber.StepDefinition 
    



      
Provides macros for defining cucumber step definitions.
Usage
defmodule AuthenticationSteps do
  use Cucumber.StepDefinition

  step "I am logged in as {string}", %{args: [username]} = context do
    {:ok, Map.put(context, :current_user, username)}
  end
end

      


      
        Summary


  
    Functions
  


    
      
        step(pattern, context_var \\ {:_, [], nil}, list)

      


        Defines a step implementation.

      step(pattern, context_var \\ {:_, [], nil}, list)


        (macro)


        
          
        

    

  


  

Defines a step implementation.

  


        

      


  

    
Gherkin.Background 
    



      
Represents a Gherkin Background section.
A Background contains steps that are run before each scenario in the feature.
It allows you to define common setup steps that apply to all scenarios.

        t()

      


    





      


      
        Types

        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %Gherkin.Background{steps: [Gherkin.Step.t()]}


      



  


        

      


  

    
Gherkin.Feature 
    



      
Represents a parsed Gherkin feature file (minimal subset).
A Feature is the top-level element in a Gherkin file, containing a name,
optional description, optional background, and one or more scenarios.
It can also have tags that apply to all scenarios in the feature.

          @type t() :: %Gherkin.Feature{
  background: Gherkin.Background.t() | nil,
  description: String.t(),
  name: String.t(),
  scenarios: [Gherkin.Scenario.t()],
  tags: [String.t()]
}


      



  


        

      


  

    
Gherkin.Parser 
    



      
Minimal Gherkin 6 parser (Feature, Background, Scenario, Step).
This module parses Gherkin feature files into Elixir structs, supporting:
	Feature with name, description, and tags
	Background with steps
	Scenarios with steps and tags
	Steps with keywords, text, docstrings, and datatables

It implements a subset of the Gherkin language focused on core BDD concepts.

    Functions
  


    
      
        parse(gherkin_string)

      


        Parses a Gherkin feature file from a string into structured data.

Parses a Gherkin feature file from a string into structured data.
This function takes a string containing Gherkin syntax and parses it into a
structured Gherkin.Feature struct with its associated components.
Parameters
	gherkin_string - A string containing Gherkin syntax

Returns
Returns a %Gherkin.Feature{} struct containing:
	name - The feature name
	description - The feature description
	tags - List of feature-level tags
	background - Background steps (if present)
	scenarios - List of scenarios

Examples
# Parse a string containing Gherkin syntax
Gherkin.Parser.parse("Feature: Shopping Cart
Scenario: Adding an item")
# Returns %Gherkin.Feature{} struct with parsed data

  


        

      


  

    
Gherkin.Scenario 
    



      
Represents a Gherkin Scenario section.
A Scenario is a concrete example that illustrates a business rule.
It consists of a name, a list of steps, and optional tags for filtering.

          @type t() :: %Gherkin.Scenario{
  name: String.t(),
  steps: [Gherkin.Step.t()],
  tags: [String.t()]
}


      



  


        

      


  

    
Gherkin.Step 
    



      
Represents a Gherkin step (Given/When/Then/And/But/*).
A Step is a single action or assertion in a scenario. It consists of:
	keyword: The step type (Given, When, Then, And, But, or *)
	text: The step text that matches step definitions
	docstring: Optional multi-line text block (triple-quoted)
	datatable: Optional table data (pipe-delimited)
	line: Line number in the source file

          @type t() :: %Gherkin.Step{
  datatable: [[String.t()]] | nil,
  docstring: String.t() | nil,
  keyword: String.t(),
  line: non_neg_integer() | nil,
  text: String.t()
}


      



  


        

      


  

    
Cucumber.StepError exception
    



      
Exception raised when a step in a Cucumber scenario fails.
This module provides detailed error reporting for two main failure cases:
	A step has no matching step definition
	A step's implementation raises an error or returns {:error, reason}

The error messages include helpful context like:
	The failing step text and location
	The scenario name and feature file
	A history of previously executed steps
	Suggestions for implementing missing steps

    Functions
  


    
      
        failed_step(step, pattern, failure_reason, feature_file, scenario_name, step_history \\ [])

      


        Creates a new step error for a step execution failure.



    


    
      
        missing_step_definition(step, feature_file, scenario_name, step_history \\ [])

      


        Creates a new step error for a missing step definition.

          @type t() :: %Cucumber.StepError{
  __exception__: true,
  failure_reason: term(),
  feature_file: String.t() | nil,
  message: String.t(),
  pattern: String.t() | nil,
  scenario_name: String.t() | nil,
  step: Gherkin.Step.t() | nil,
  step_history: list()
}


      



  


        

      

      
        Functions

        


    

  
    
      
    
    
      failed_step(step, pattern, failure_reason, feature_file, scenario_name, step_history \\ [])



        
          
        

    

  


  

Creates a new step error for a step execution failure.
When a step implementation raises an exception or returns {:error, reason},
this function creates a detailed error message with context about the failure.
Parameters
	step - The Gherkin.Step struct that failed during execution
	pattern - The step pattern that matched the step
	failure_reason - The exception or error reason
	feature_file - The path to the feature file containing the step
	scenario_name - The name of the scenario containing the step
	step_history - A list of previously executed steps with their status

Returns
Returns a Cucumber.StepError struct with a formatted error message and context.
Examples
The error message will include details like:
Step failed:

  Then the result should be 10

in scenario "Basic math" (test/features/calculator.feature:12)
matching pattern: "the result should be {int}"

Expected 10 but got 9

Step execution history:
  [passed] Given I have entered 5 into the calculator
  [passed] And I have entered 4 into the calculator
  [passed] When I press add
  [failed] Then the result should be 10

  



    

  
    
      
    
    
      missing_step_definition(step, feature_file, scenario_name, step_history \\ [])



        
          
        

    

  


  

Creates a new step error for a missing step definition.
When a step in a feature file doesn't match any step definition, this function
creates a helpful error message that includes a suggestion for implementing
the missing step.
Parameters
	step - The Gherkin.Step struct that has no matching definition
	feature_file - The path to the feature file containing the step
	scenario_name - The name of the scenario containing the step
	step_history - A list of previously executed steps with their status

Returns
Returns a Cucumber.StepError struct with a formatted error message and context.
Examples
The error message will include a suggested implementation like:
step "I click the submit button", context do
  # Your step implementation here
  context
end

